
News  

Travel disruption due to volcanic activity 
On Wednesday 14th April the Eyjafjallajoekull volcano in Iceland erupted, creating a cloud 
of ash and debris that has since spread to European airspace and caused the 
cancellation of flights to and from the UK and much of Europe. Although some restrictions 
on flights to and from European airports have now been lifted the major UK airports 
remain closed at this time and, due to these travel restrictions, we regret that we will be 
unable to attend the following meetings:  

6th European Conference on Comparative Neurobiology 
Science Museum Principe Felipe, 
Valencia, Spain 
April 21st – 23rd 2010 

FASEB Experimental Biology 2010 
Anaheim Convention Center, 
Anaheim, CA, USA 
April 24th – 28th 2010 

Our sincere apologies to those people who were planning to see us at these meetings 
and to those who had arranged to meet with us in their laboratories during these visits. 
The widespread cancellation of flights leaving the UK has also introduced a delay to our 
deliveries of recent orders, but we remain hopeful that the situation will improve in the 
next few days.   

Future meetings and events 

UCL Neuroscience Symposium 2010 
Senate House, University of London 
London, UK 
April 29th 2010 

10th International Basal Ganglia Society Meeting 
Ocean Place Resort, 
Long Branch, NJ, USA 
June 20th – 24th 2010 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 7.03 Signal version 4.07 

Spike2 version 6.14 Signal version 3.12 

Spike2 demo Signal demo 

Spike2   

Q. It would be very nice if Y-axis units in Spike2 could automatically scale when 
zooming in to view a small area of interest. 

A. From version 7.03, the X and Y axes in Spike2 can be set to automatically scale 
the units by powers of 1000 and can be displayed using SI scaling prefixes 
(M,k,m,u n,… etc.). This option is available in the X and Y Range dialogs. The 
display change is purely cosmetic so while values displayed on screen by 
horizontal cursors and interactive measurements (made by holding down the Alt 
key and clicking and dragging on a channel) will take the Auto-units setting into 
account, this does not affect the actual values associated with the axis as seen 
by the Spike2 program or by the script language. 

 
Y-axis display showing the different auto-units display 

 modes on duplicated channels 
 

 

  

Q. I would like to be able to monitor a waveform channel to trigger sampling in 
response to a level crossing. Are there options available for this in the standard 
sampling configuration? 

A. Signal can use the built-in Peri-trigger sweep mode to trigger pulse outputs 
based on digital and event markers or threshold crossings on a sampled 
waveform channel. The highest numbered channel in the data file is always used 
to monitor triggering based on a waveform input. The trigger type and pre-trigger 
time of data to display can be set in the Peri-trigger tab of the sampling 
configuration, which is enabled when the Peri-trigger sweep mode is selected in 
the General tab.  

 
Peri-trigger settings and configuration using a rising threshold trigger 

CED
The Auto-units option is only available if the Logarithmic axis display option is un-checked

CED
You can set a negative value in the Pre-trig. Time field to discard data points in a time range after the trigger has occurred

CED
Signal always monitors the highest numbered channel in the file for the analogue trigger channel. You can set which ADC port you want to use for this channel by including it at the end of the ADC ports list in the General tab. In this example, ADC port 0, which corresponds to channel 4, will be used



 For triggering based on a waveform channel you can select between rising 
threshold (Analogue +), falling threshold (Analogue -) or outside a pair of 
thresholds (Analogue =). You can set a default threshold value and a hysteresis 
level to avoid triggering from noise in the waveform. When using a rising 
threshold, the hysteresis value sets a level that the signal must fall below before 
being able to trigger the threshold again. 

You can adjust the threshold and hysteresis level during sampling by clicking and 
dragging the horizontal cursors on the highest numbered data channel in the file. 

 

Scripts: Spike2  

Q. I want to generate a sinusoid stimulus with a frequency that increases linearly. Is 
this possible using a script and sequence? Ideally I would also like to have the 
option of generating a quadratic increase in frequency as well.  

A. Following the above request we have written an example script 
ZAPexample.s2s, and associated sequence file, ZAPexample.pls, which can 
be used with Spike2 version 7 to generate a sine wave with a linear or quadratic 
increasing frequency from DAC 0. To view the output sine wave, set up a 
waveform channel with a sample rate of 10kHz, connect DAC 0 to the ADC port 
required for this channel and load the sequencer file into your sampling 
configuration before running the script. The script opens a new file ready to 
sample based on the current sampling configuration. Sine wave settings are 
controlled from a script toolbar button and can be updated during sampling.  

Scripts: Signal  
Q. Is there any way to replay a Signal data file? This would be very useful for 

presentations. 

A. The example script, Rerun.sgs, can be used to replay a Signal data file, along 
with any associated memory views or trend plots, on a frame by frame basis. 
Each frame is overdrawn to show the changes in data over the course of the 
recording. 

Did you know…? 
The maximum number of horizontal cursors available has been increased from 4 to 10 in 
Signal and from 4 to 9 in Spike2. We are reserving a special horizontal cursor 0 in Spike2 
for future use. 

Recent questions   
Q. I am interested in creating a spectrogram to show the amplitude of the different 

frequencies in a waveform, so the X axis would show time and the Y axis would 
show frequency content. 

A. You can display the frequency content of a waveform over time using the 
Sonogram drawing mode. Right-click in the data area of the waveform channel 
and select the Channel Draw mode… option and then select Sonogram from the 
drop-down list. This enables some other fields in the dialog that you can use to 
set Block size, window type and range. Clicking Draw will re-draw the waveform 
as a sonogram. The default display is Monochrome, but you can select different 
colour layouts and customised colour maps by right-clicking on the sonogram 
channel and selecting the Sonogram Colours… option from the context menu.  

  
Sonogram display of a speech waveform  

using the Thermal colour map 

'$ZAPexample|Script that plays out a sine wave of frequency related to a functioon of time
'First attempt using a linear frequency ramp
'This script should be used in conjunction with the ZAPexample.pls sequence file

'It contains toolbar to open a new data file for sampling and
'start and stop sampling. It displays the log view at the bottom
'of the screen to place values in that the user may wish to extract.
'Hides/restores unwanted windows. Manages data files (one file at a time).

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 13/04/10

var data%;                      'Handle of new data file (Spike2 sets it to 0)
var sTime;                      'The last time we looked at the idle routine
var sweepType% := 0;                                        'Method of frequency ramping (linear or quadratic)
var quadA := 0;
var quadB := 0.02;
var startFreq:=0.1;            'Default output settings
var endFreq:=50;
var length:=30;
var ok%;
var init%;
var inc;
var incVar%;
var steps%;
var numEvents%;
var int%;						'Array element of current interval	
var ind%;

var freq[10000];				'Array of frequencies
var freqVar%[10000];
var int[10000];					'Array of intervals
var tInt[10000];				'Temporary array of intervals
var stimTime;					'Time of stimulus relative to start of ramp
var ticks%;
var sineAmp:= 1;                'Sine wave amplitude (0 - 1)
var sineAmp%;                   'Amplitude as sequencer variable

var qFIndex%;                   'Dialog index number for quadratic equation text
var mfIndex%;
var mFreqText$ :="                      ";                 'Text to show maximum frequency for quadratic sweep
var mFreqIndex%;



HideAll();                      'Hide all unwanted bars
ToolbarVisible(1);              'Make toolbar visible always
New%();                         'Set up new sampling window
DoToolbar();                    'Do the toolbar
RestoreAll();                   'Restore all bars
Halt;

func New%()                     'New sampling window
'View(LogHandle());              'Make log view the current view
'EditSelectAll();                'Select all text in log view
'EditClear();                    'Delete it
'Window(0,80,100,100);           'Display it at the bottom of the screen
'WindowVisible(1);               'Make it visible

if data%>0 then                 'If there is already a data view open...
  View(data%);                  '...close it
  FileClose();                  'No point setting data% to 0...
endif;

data%:=FileNew(0,1);            'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
DrawMode(-1,2);                 'Set event draw mode to lines
Window(0,0,100,100);             'Make data window in top bit of screen
XRange(0,10);

'FrontView(LogHandle());         'Bring the Log view to the front
FrontView(data%);               'Bring the data view to the front
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(2,1);             'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);    'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarSet(6,"&Wave Settings", OutWaveSet%);
ToolbarSet(8,"&Output Wave",DoWave%);
ToolbarEnable(3,0);             'Disable "Sample stop" button 
ToolbarEnable(8,0);
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()                    'If "Quit" is pressed
SampleStop();                   'Stop sampling
return 0;                       'leave toolbar
end;

func Start%()                   'If "Start" is pressed
sTime := 0;                     'set start of anaylsis time
SampleStart();                  'Start sampling
ToolbarEnable(4,0);             'Disable "New file" button
ToolbarEnable(3,1);             'Enable "Sample stop" button
ToolbarEnable(2,0);             'Disable "Sample start" button
ToolbarEnable(1,0);             'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;                       'Stay with toolbar
end;

func Stop%()                    'If "Stop" is pressed
SampleStop();                   'Stop sampling
if (data%>0) and                'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);             'Enable "New file" button
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(1,1);             'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;                       'Stay in toolbar
end;

Func OutWaveSet%()
var i%;
var sectionEvents%:= 0;
var c%;
var freqStep;

'Message("Count = %d\nInd = %d\nInt = %d",SampleSeqVar(2),SampleSeqVar(6),SampleSeqVar(1));  used for sequencer debug

DlgCreate("Output wave settings");
DlgList(1,"Frequency sweep type","Linear|Quadratic",2,30);
qFIndex% := DlgText("f(t) = f0 + at + bt^2",55,1);
DlgReal(2,"Initial Frequency (f0)",0.1,100,30,2);
DlgReal(3,"End Frequency",0.1,1000,30,3);
DlgReal(4, "Amplitude 0 - 1 (1=+/-5V)",0,1,30,5);
DlgReal(5,"Length (s)",1,60,30,6);
DlgReal(6,"a",0,1000,30,3);
DlgReal(7,"b",0,100,30,4);
mfIndex% := DlgText("Maximum frequency",55,3);
mFreqIndex% := DlgText(mFreqText$,58,4);
DlgAllow(1023,0,Change%);
ok%:=DlgShow(sweepType%,startFreq,endFreq,sineAmp,length,quadA,quadB);             'Get wave output settings

if ok% > 0 then
    sectionEvents% := length*10000/5;                       'Get number of frequency changes during output (0.5ms per change)
	Resize freq[sectionEvents%];                            'Resize arrays to correct size
    Resize freqVar%[sectionEvents%];
    if sweepType% = 0 then
        freqStep := (endFreq-startFreq)/sectionEvents%;         'Get frequency step per change
        freq[0] := startFreq;                                   'Set start frequency
        ArrConst(freq[1:sectionEvents%-1],freqStep);            'Copy frequency step to array
        ArrIntgl(freq[]);                                       'Integrate to form frequency ramp
    else
        freq[0] := 0;
        ArrConst(freq[1:sectionEvents%-1],0.0005);
        ArrIntgl(freq[]);
        for i% := 0 to sectionEvents%-1 do
            freq[i%] := GetQuadFreq(startFreq,freq[i%],quadA,quadB);
        next;
    endif;
    for c%:= 1 to sectionEvents% do
        freqVar%[c%-1]:=vhz%(freq[c%-1],10);                'Convert frequencies to integer vales to sequencer
    next;
    SampleSeqTable(freqVar%[:sectionEvents%]);              'Write freq. values to table
    init%:= vhz%(startFreq,10);                             'Get initial frequency value
    SampleSeqVar(2,sectionEvents%);
    SampleSeqVar(3,init%);                                  'Write to sequencer
    sineAmp% := sineAmp * 32768;
    SampleSeqVar(8,sineAmp%);                               'Set sine wave amplitude
    ToolbarEnable(8,1);                                     'Enable the output button
endif
return 1;
end;

Func Change%(n%)
var mFreq;
if DlgValue(1) = 0 then
    DlgEnable(1,3);
    DlgVisible(1,3);
    DlgVisible(0,qFIndex%);
    DlgEnable(0,6);
    DlgVisible(0,6);
    DlgEnable(0,7);
    DlgVisible(0,7);
    DlgVisible(0,mfIndex%);
    DlgVisible(0,mFreqIndex%);
else
    DlgEnable(0,3);
    DlgVisible(0,3);
    DlgVisible(1,qFIndex%);
    DlgEnable(1,6);
    DlgVisible(1,6);
    DlgEnable(1,7);
    DlgVisible(1,7);
    mFreq := GetQuadFreq(DlgValue(2),DlgValue(5),DlgValue(6),DlgValue(7));
    mFreqText$ := Print$("%.4f Hz",mFreq);
    DlgValue$(mFreqIndex%,mFreqText$);
    DlgVisible(1,mfIndex%);
    DlgVisible(1,mFreqIndex%);
endif;
return 1;
end;

Func GetQuadFreq(sFreq,t,a,b);
var cFreq;
cFreq := sFreq + (a*t) + (b*(t*t));
return cFreq;
end;



Func DoWave%()
SampleKey("g");         'Start output wave
return 1;
end;

'You can find this code in the App() on-line help.
var gFloat%[20];                'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);          'number of windows
for i% := 1 to gFloat%[0] do    'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do    'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
'
' The vhz% function converts a cosine frequency in Hz into an integer value
'  (the result of the function) suitable for passing into a sequencer variable
'  for use in the RATE, RATEW, RINC and RINCW sequencer instructions. It mimics
'  the vhz() expression in the text sequence compiler.
'
' hz      is the frequency in Hz to be converted.
' tick    is the sequencer tick period in milliseconds. This is exactly the same
'         as the msPerStep field in the SET and SCLK directives.
'
func vhz%(hz, tick)
return vangle%(hz * 0.36 / tick);
end

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
'
' The vangle% function converts a value in degrees into an integer value (the
'  result of the function) suitable for passing into a sequencer variable for
'  use in the ANGLE and PHASE cosine control sequencer instructions. It mimics
'  the vangle() expression in the text sequence compiler.
'
' It mimics the vangle expression in the text sequence compiler to make it
'  easier to calculate sequencer variable values in your script.
'
' deg     is the phase angle in degrees to be converted.
'
func vangle%(deg)
deg := deg mod 360;                     ' First of all handle stupid values
if deg >= 180 then deg := deg-360 endif;' Then adjust to make the arithmetic easier
if deg < -180 then deg := deg+360 endif;
var v32;
v32 := deg * 2147483648.0 / 180.0;      ' This result must be in integer range
var v32%;
if v32 > 2147483647.0 then              ' 2147483647 is the largest positive
    v32% := 2147483647;                 '  integer value
else
    if v32 < -2147483647.0 then         ' -2147483648 is the lowest negative value
        v32% := -2147483647;            '  but we have to stay one above that
    else
        v32% := round(v32);             ' We can be sure this will not overflow
    endif;
endif;
return v32%;                            ' So we can return as an integer
end


CED
ZAPexample.s2s




CED
ZAPexample.pls


'$Rerun.sgs|Example script to re-run a data file along with any associated memory views
'and trend plots

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 19/04/10

var data%;          'View handle for data file
var f%;        'Frame counter
var res%[10];        'Array to hold associated result views
var i%;               'Counter for result views
var sFrame%:=1;
var eFrame%;         'Start and end frame number

if ViewKind() = 0 then              'Check we have a data file open
data%:= View();                      'If so, use it
FrontView(data%);
else
data%:= FileOpen("",0);        'If not prompt user to select one
WindowVisible(1);
endif;
if data%<0 then Message("Unable to open new data file");Halt; endif;       

ViewList(res%[],16+4096+8192);     'Get a list of any associated result views

DoToolbar();  

Func DoToolbar();
ToolbarSet(1,"Quit");
ToolbarSet(2,"&Re-run",ReRun%);       'Re-run function
return toolbar("Re-run script",1023);
end;

Func ReRun%()
f%:=1;           'Set frame counter to 1
Repeat
	View(data%);                'Make sure data is the current view
	Frame(f%);            'Show current frame
    eFrame% := f%;
    OverDrawFrames(sFrame%,eFrame%,-1,0);   'Overdraw all frames from first to current
    OverDraw(1,6);
	for i% := 1 to res%[0] do           'For any associated result views
		if f% = 1 then
			View(res%[i%]).Process(View(-1).MinTime(),1,1);     'Process the view
		else
			View(res%[i%]).Process(View(-1).MinTime(),0,1);
		endif;
        next;
    Yield(0.5,1023);      'Yield to give update time between frames
	f% += 1;         'Increment frame counter
	
until f% > FrameCount();
return 1;
end;

CED
Rerun.sgs



User group      
Try the CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

